Interpolazione Lagrange/Newton/Esempi

(lezione 8 aprile 2013)

```
> restart ;
> with(plots):
```

Esempio semplice, confronto Lagrange e Newton

Tabella di punti di interpolazione

```
> X := [-1,-2, 0, 3, 6, 7, 8] :
  Y := [ 0, 0, 1, 2, 5, 1, 1] :
  N := 6 ;
```

$$N := 6 \tag{1.1}$$

Costruzione polinomi di Lagrange

```
> Lagra := proc ( N, k, X )
    local i, res;
    res := 1 ;
    for i from 0 to N do
      if i <> k then
        res := res * (x-X[i+1])/(X[k+1]-X[i+1]) ;
      end;
    end;
    res ;
  end proc :
> for k from 0 to N do
    L||k := expand(Lagra( N, k, X )) ;
  end ;
```

$$L0 := -\frac{1}{2016}x^6 + \frac{11}{1008}x^5 - \frac{23}{288}x^4 + \frac{89}{504}x^3 + \frac{15}{56}x^2 - x$$

$$L1 := \frac{1}{7200}x^6 - \frac{23}{7200}x^5 + \frac{37}{1440}x^4 - \frac{113}{1440}x^3 + \frac{13}{400}x^2 + \frac{7}{50}x$$

$$L2 := \frac{1}{2016}x^6 - \frac{1}{96}x^5 + \frac{139}{2016}x^4 - \frac{65}{672}x^3 - \frac{4}{9}x^2 + \frac{41}{56}x + 1$$

$$L3 := -\frac{1}{3600}x^6 + \frac{1}{200}x^5 - \frac{17}{720}x^4 - \frac{1}{60}x^3 + \frac{179}{900}x^2 + \frac{14}{75}x$$

$$L4 := \frac{1}{2016}x^6 - \frac{5}{672}x^5 + \frac{7}{288}x^4 + \frac{11}{224}x^3 - \frac{151}{1008}x^2 - \frac{1}{6}x$$

$$L5 := -\frac{1}{2016}x^6 + \frac{1}{144}x^5 - \frac{41}{2016}x^4 - \frac{23}{504}x^3 + \frac{1}{8}x^2 + \frac{1}{7}x$$

$$L6 := \frac{1}{7200}x^6 - \frac{13}{7200}x^5 + \frac{7}{1440}x^4 + \frac{17}{1440}x^3 - \frac{3}{100}x^2 - \frac{7}{200}x \tag{1.2}$$

```
> A := plot( [seq([X[k+1],0],k=0..N)], style=point, symbol=
  solidcircle,symbolsize=20 ):
> B := plot( [seq([X[k+1],1],k=0..N)], color=blue,style=point,
  symbol=solidcircle,symbolsize=10 ):
```

```
> C := plot( [L||(0..N)], x=-4..9, y=-2..2 ):
> display(A,B,C);
```


2
1
y
−1
−4
−2
0
2
4
6
8
x

Calcolo polinomio interpolate con formula di Lagrange

```
> p := add( Y[k+1]*L||k, k=1..N ) ;
```

$$p := 1 + \frac{19}{50}x - \frac{17657}{25200}x^2 + \frac{823}{10080}x^3 + \frac{23}{180}x^4 - \frac{1637}{50400}x^5 + \frac{13}{6300}x^6 \qquad (1.3)$$

```
> A := plot( [seq([X[k+1],Y[k+1]],k=0..N)], style=point, symbol=
  solidcircle,symbolsize=20 ):
> B := plot( p, x=-4..9, y=-2..8 ):
> display(A,B);
```



Costruzione polinomi di Newton

```
> Omega := proc ( k, X )
    local i, res;
    res := 1 ;
    for i from 0 to k-1 do
      res := res * (x-X[i+1]) ;
    end;
    res ;
  end proc :
> for k from 0 to N do
    O||k := expand(Omega( k, X )) ;
  end ;
```

$$O0 := 1$$

$$O1 := x + 1$$

$$O2 := x^2 + 3x + 2$$

$$O3 := x^3 + 3x^2 + 2x$$

$$O4 := x^4 - 7x^2 - 6x$$

$$O5 := x^5 - 6x^4 - 7x^3 + 36x^2 + 36x$$

$$O6 := x^6 - 13x^5 + 35x^4 + 85x^3 - 216x^2 - 252x \quad \textbf{(1.4)}$$

```
> evalDD := proc( N, X, Y )
    local i, k, DD ;
    DD := Y ;
    for k from 1 to N do
      for i from N to k by -1 do
        DD[i+1] := (DD[i+1]-DD[i])/(X[i+1]-X[i+1-k]):
      end:
    end:
    DD ;
  end proc :
```

```
> DD := evalDD(N,X,Y) ;
```

$$DD := \left[0, 0, \frac{1}{2}, -\frac{2}{15}, \frac{109}{5040}, -\frac{19}{3360}, \frac{13}{6300}\right] \quad \textbf{(1.5)}$$

Vediamo i polinomi intermedi

```
> for i from 1 to N do
    P||i := add( DD[k+1]*O||k, k=1..i ) ;
  end :
```

```
> A := plot( [seq([X[k+1],Y[k+1]],k=0..N)], style=point, symbol=
  solidcircle,symbolsize=20 ):
```

```
> B := plot( [P2,P3,P4], x=-4..9, y=-2..8 ):
  C := plot( [P5,P6], x=-4..9, y=-2..8 ):
```

```
> display(A,B);
  display(A,C);
```


8
6
4
2
−2
y
−4
−2
0
2
4
6
8
x


8
6
4
2
−2
y
−4
−2
0
2
4
6
8
x